

Paper 157-25

Want Quick Results? An Introduction to SAS/GRAPH® Software

Arthur L. Carpenter
California Occidental Consultants

ABSTRACT

SAS/GRAPH software contains the tools and flexibility to produce high-resolution presentation quality graphs. With this flexibility comes a complexity that often discourages a new user from taking the plunge and learning SAS/GRAPH. Learning and using SAS/GRAPH does not need to be difficult. Nor are the more complex aspects of SAS/GRAPH always needed to produce graphic output that is just what your boss wants.

This workshop will introduce three of the primary procedures used by SAS/GRAPH, and introduce some of the options and statements that make them special. We will start with TITLES and FOOTNOTES. PROC GPLOT will be used to create a scatterplot which will be enhanced with SYMBOL statements and a number of options. A couple of variations of GCHART will be demonstrated and the creation of title slides using GSLIDE will be discussed briefly. Options specific to SAS/GRAPH and how and when they should be used will be handled with an introduction to the GOPTIONS statement.

KEY WORDS

GOPTIONS, GPLOT, GCHART, SYMBOL, AXIS, TITLE, FOOTNOTE, OPTIONS

GRAPHICS OPTIONS

Options specific to the graphics environment are specified using the GOPTIONS statement in much the same way as the OPTIONS statement is used to specify system options. Typical GOPTION statements are shown in the following code. Remember SAS® is not case sensitive in open code.

```
goptions reset=all;
GOPTIONS CBACK=blue DEVICE=win
  prompt noborder;
GOPTIONS ftext=swiss colors=(white)
  ctext=white;
goptions htext=2;
```

As in the OPTION statement one or more options can be included in the GOPTIONS statement. In the above code all four SAS statements could have been written in a single statement. I like to group related options in their own statement so that I can comment out the ones that I don't want to apply.

The options shown above include:

reset=all	all goptions are reset to their default values
device=win	determines the type of device for graphics, in this case windows
cback	set background color
ftext, ctext, and htext	set the font, color and size of the text

The RESET= graphics option can also be used to reset specific groups of settings to their defaults e.g. `reset=symbols` resets any symbol statement definitions.

TITLES AND FOOTNOTES

SAS/GRAPH procedures also use TITLE and FOOTNOTE statements. Unlike in Base SAS titles and footnotes, however, options are available to change the color, font, size, justification, and orientation of the text. The option is applied to the text string following the option. Options (with examples in this paper) include:

- text color
c=blue
- size of the text
h=2
- font selection
f=simplex
- angle (of text string)
a=90


- rotation (angle of individual characters)
r=45
- justification of the string
j=center

In Figure 1 three titles are defined and presented using PROC GSLIDE.

```
proc gslide;
title1 'Figure 01';
title2 h=3 font=swissb 'swissb'
      color=yellow f=cartog
      'J K L M N';
title3 h=1 f=simplex
      'H1' h=2 'H2' h=4 'H4';
run;
```

Notice that options can be redefined within a given TITLE or FOOTNOTE.

Figure 1



PROC GSLIDE is used to display titles, notes, and footnotes, and does not use data. This procedure is especially useful when you want to produce pictures, drawings, or title slides for a presentation.

USING PROC GPLOT

The syntax for PROC GPLOT is very similar to that used in PROC PLOT. Typical syntax will include the following statements:

```
proc gplot options;
plot yvar * xvar / options;
run;
```

where the PLOT statement defines the plot definition by specifying the vertical (yvar) and horizontal (xvar) variables and any options

associated with the axes. PLOT statement options might include:

HREF & VREF

Used to define vertical and horizontal reference lines. The following statement causes vertical lines to be drawn at the values of x=2, 4, and 6. HREF creates vertical lines and these are tied to the noted values on the horizontal axis.

```
plot y*x/href=2 4 6;
```

HAXIS & VAXIS

Used to assign options and control parameters to an axis. The first of the two statements below constrains the horizontal axis to the values between 10 and 100 and causes a major tick mark at 10, 20, 30, ... The second indicates that the horizontal axis will be controlled by an AXIS statement.

```
plot y*x/haxis=10 to 100 by 10;
plot y*x/haxis=axis1;
```

When specifying values to the axis and tick marks the syntax is very similar to that used in DO statements.

The following code includes a GPLOT that utilizes TITLES and a FOOTNOTE as well as axis control options. The variable defining the vertical axis is BMCRUS while DATE is the horizontal plot variable. The HAXIS= option instructs GPLOT to make major tick marks on the first of each month between 01Jun85 and 01Sep85.

```
proc gplot data=quick.biomass;
plot bmcрус * date /
      haxis='01jun85'd to
           '01sep85'd by month;
title1 'Crustacean Biomass';
title2 a=90 h=1.5
      'grams wet weight';
footnote1 j=1 h=1.5 f=simplex
         'Figure 2';
run;
```

TITLE2 uses the a= option to rotate the text line from horizontal to vertical (along the left side).




Figure 2

In Figure 2 the label for the vertical variable (BMCRUS) is redundant with TITLE2 and should be removed (see Figure 3). Control over the axis labels and other aspects of the axis can most easily be accomplished through the use of AXIS statements.

AXIS STATEMENTS

The AXIS statement allows you to maintain very specific control for all aspects for each individual axis in the plot. You may have up to 99 AXIS statements defined at any given time, although due to performance considerations it is usually wiser to only define those that will be used during the current execution.

```
axis<1...99> options=optionvalues;
```

Axis options can be grouped into four primary areas according to how they effect the axis. The four areas and some sample options include:

- axis scale options control axis scaling. Those below use log scales.


```
logbase=10
logstyle=expand
```
- appearance options are used to change text and line characteristics such as color, font, size and line type.


```
color=black
style=3
```
- tick mark options specify the characteristics of the tick marks. This includes removing tick marks altogether or

changing their frequency, size, or color.

```
major=none
major=(h=1.5 c=blue)
minor=(n=3 h=1.0 c=blue)
```

- text options control text associated with tick marks (VALUE) or axis labels (LABEL). Most of the options used with TITLES can also be used here.

```
label=('mean weight in grams')
value=(a=45 h=1.5)
```

The following AXIS statement requests LOG scaling using base 10, and eight minor tick marks between each major tick mark. Minor tick marks will have a length of 1 and major ticks will be half again as large. Simplex will be used for the tick mark text font, and the label will not be displayed.

```
axis1 logstyle=expand logbase=10
      minor = (n=8 h=1)
      major = (h=1.5)
      value = (f=simplex h=1.5)
      label = none;
```

In Figure 2 the HAXIS= option was used to control the date tick marks. The following AXIS statement accomplishes the same thing using the ORDER= option.

```
axis2 order = '01jun85'd to
             '01sep85'd by month
      value = (f=simplex h=1.5);
```

These two AXIS statements can be used together with the following code to redisplay Figure 2 and create Figure 3.

```
proc gplot data=quick.biomass;
plot bmcrus * date / vaxis=axis1
                    haxis=axis2;
symbol v='*' f=simplex;
title1 'Crustacean Biomass';
title2 a=90 h=1.5 'grams wet weight';
footnote1 j=1 h=1.5 f=simplex
          'Figure 3';

run;
```

Notice that with the expanded vertical axis, a great deal of detail lost in Figure 2 becomes apparent.




Figure 3

SYMBOL STATEMENTS

The SYMBOL statement provides control over the characters to be printed on the scatter plot.

```
symbol<1...99> option=optionvalue;
```

Symbol options are used to control various aspects of the symbol. Text control options are similar to those used in TITLE and AXIS statements. $h=$, $f=$, and $c=$ apply to the height, font, and color respectively. Other useful SYMBOL statement options include:

- $value$ is used to designate the plot symbol.


```
v=star  
v='*'
```
- $line$ is used to designate the type of line ($line=1$ is solid while $line=2$ is dashed)


```
l=33
```
- $interpolation$ determines the method used to connect plot points


```
i=box
```

Like the options in the TITLE and FOOTNOTE statements, most of the options can be abbreviated by their first letter.

The following SYMBOL and AXIS statements were used to produce Figure 4. The font ($f=$), plot character ($v=$), and its size ($h=$) are selected. A solid line ($l=$) is used to join ($i=$) the plot points.

```
axis1 label=(h=1.5 a=90 f=swiss 'V'  
            f=simplex 'olume (X1000)');  
axis2 order=('03aug81'd to  
            '31aug81'd by week)  
value=(h=1.5);  
  
proc gplot data=quick.dow;  
plot volume*date=1 / vaxis=axis1  
                    haxis=axis2;  
symbol1 v='V' h=2 f=simplex l=1  
        I=join c=black;  
format date date7.;  
title1  
      'Stock Market Analysis of the Dow';  
title2 h=2 f=simplex 'Daily Volume';  
note h=1.5 move=(20,25)pct 'Volumes';  
  
footnote1 j=1 h=1.5 f=simplex  
         'Figure 4';  
  
run;
```




Figure 4

The vertical axis label for this figure is supplied through the AXIS1 statement rather than with the TITLE statement as in Figure 3. The NOTE statement is used to write the word 'Volumes' directly under the line. It is placed on the graph manually using the MOVE= option. The major tick marks for the horizontal axis are dates and are specified using BY WEEK in the AXIS2 ORDER= option.

USING PROC GCHART

The general syntax for using PROC GCHART is shown below. The primary statement that you need to be able to use is the one that selects the type of chart you wish to create.

```
proc gchart data=datasetname;  
charttype varlist / options;
```

```
run;
```

The possible types of charts (*charttype*) include:

- vbar vertical histograms
- hbar horizontal histograms
- block tower or block chart
- pie pie diagram
- star star or spine chart
- donut donut charts

Each type of chart has its own options, however several apply to most of the charts. These include:

- sumvar summary variable (numeric)
- type summary statistic
- midpoints specific midpoint values
- discrete numeric variable to be treated as categorical

The following code creates Figure 5 as a vertical histogram (VBAR). The variable DATE is used to define vertical bars. Since the summary statistic (TYPE) was specified as MEAN, the average total biomass (BMTOTL) for the date will be displayed.

```
axis1 value=(h=1.5 f=simplex a=45 r=0)
      label=(h=1.5 f=simplex);

proc gchart data=biomass;
vbar date / discrete sumvar=bmtotl
      type=mean maxis=axis1;
title1 'Average Total Biomass';
footnote1 j=1 h=1.5 f=simplex
      'Figure 5';
run;
```

The AXIS1 statement is tied to the horizontal or midpoint axis through the use of the MAXIS option. This statement changes the orientation of the labels of the bars (DATES). Notice the use of R=0, this should not be needed but is sometimes required with the A= option.




Figure 5

The label for the vertical axis is coming from the label for the summary variable (BMTOTL) and looks messy because SAS has selected a hardware font. There are several things that you can do to control this label. To avoid defaulting to a hardware font, be sure to always specify the font of your choice. In this case a simple GOPTIONS statement would have done the trick.

```
Goptions ftext=simplex;
```

You can also control this label as was done in Figures 2 & 3 by using a title with the A= option, or by using an AXIS statement with the response axis as is done in Figures 4 & 6.

The following code also creates a vertical histogram with the variable BMTOTL being used to define vertical bars. Since no summary statistic (TYPE) was specified the frequency (number of observations falling into each group) is displayed.

Unless directed otherwise GCHART will determine the midpoints or size of the groupings used to determine the bars. The MIDPOINT option is used to explicitly define the groups, one of which, in this example, has an extreme value (30).

It is possible to use AXIS statements on the response axis (RAXIS=), the midpoint axis (MAXIS=), and even the grouping axis (GAXIS=).

```
axis1 value=(h=1.5 f=simplex)
      label=none;

axis2 value=(h=1.5 f=simplex)
      label=(h=1.5 f=simplex);
```

```

* one observation has a large value
* of BMTOTL;
proc gchart data=quick.biomass;
vbar bmtotl /
midpoints=.5 to 6 by .5, 30
raxis=axis1
maxis=axis2;
title1 'Total Biomass Frequency';
title2 h=1.5 f=simplex
'Using uneven midpoints';
title3 h=1.5 f=simplex a=90
'Frequency';
footnote1 j=1 h=1.5 f=simplex
'Figure 6';
run;

```




Figure 6

Figure 6

It should be noted that when using midpoints with unequal interval widths the values noted are not necessarily the true midpoint of the interval. In the above example the interval labeled as 6.0 ranges from 5.5 to 18 while the interval labeled 30.0 ranges from 19 and up.

SUMMARY

This has been a very brief introduction to some of the features of SAS/GRAPH. Although only a very few of the available options are included here, even these are sufficient to produce respectable graphs and charts rather quickly. Refinements are available in the form of many more options and supporting statements. These additional options and statements are further described in *Quick Results with SAS/GRAPH® Software* and in the book by Thomas Miron, *The How-To Book for SAS/GRAPH® Software*.

AUTHOR INFORMATION

Art Carpenter's publications list includes three books (*Annotate: Simply the Basics*, *Quick Results with SAS/GRAPH® Software*, and *Carpenter's Complete Guide to the SAS® Macro Language*), two chapters in *Reporting from the Field*, and over three dozen papers and posters presented at SUGI, PharmaSUG, and WUSS. He has been using SAS since 1976 and has served in a variety of positions in user groups at the local, regional, and national levels.



Art is a SAS Quality Partner™ and SAS Certified Professional™. Through California Occidental Consultants he teaches SAS courses and provides contract SAS programming support nationwide.



AUTHOR CONTACT

Art Carpenter
California Occidental Consultants
P.O. Box 6199
Oceanside, CA 92058-6199

(760) 945-0613

art@caloxy.com
www.caloxy.com

REFERENCES

Carpenter, Arthur L. and Charles E. Shipp, *Quick Results with SAS/GRAPH® Software*, Cary, NC: SAS Institute Inc., 1995. 249 pp.

Miron, Thomas, *The How-To Book for SAS/GRAPH Software*, Cary, NC: SAS Institute Inc., 1995. 286 pp.

TRADEMARK INFORMATION

SAS, SAS/GRAPH, SAS Certified Professional, and SAS Quality Partner are registered trademarks of SAS Institute, Inc. in the USA and other countries.

® indicates USA registration.